

NAG Library Routine Document

G05YKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05YKF generates a quasi-random sequence from a log-normal distribution. It must be preceded by a call to one of the initialization routines G05YLF or G05YNF.

2 Specification

```
SUBROUTINE G05YKF(XMEAN, STD, N, QUASI, IREF, IFAIL)
  INTEGER          N, IREF(liref), IFAIL
  double precision XMEAN(idim), STD(idim), QUASI(N,idim)
```

3 Description

G05YKF generates a quasi-random sequence from a log-normal distribution by first generating a uniform quasi-random sequence which is then transformed into a log-normal sequence using the exponential of the inverse of the Normal CDF. The type of uniform sequence used depends on the initialization routine called and can include the low-discrepancy sequences proposed by Sobol, Faure or Niederreiter. If the initialization routine G05YNF was used then the underlying uniform sequence is first scrambled prior to being transformed (see Section 3 in G05YNF for details).

4 References

Brately P and Fox B L (1988) Algorithm 659: Implementing Sobol's Quasirandom Sequence Generator *ACM Trans. Math. Software* **14** (1) 88–100

Fox B L (1986) Algorithm 647: Implementation and Relative Efficiency of Quasirandom Sequence Generators *ACM Trans. Math. Software* **12** (4) 362–376

Wichura (1988) Algorithm AS 241: The percentage points of the Normal distribution *Appl. Statist.* **37** 477–484

5 Parameters

Note: the following variables are used in the parameter descriptions:

idim = IDIM, the number of dimensions required, see G05YLF or G05YNF;

liref = LIREF, the length of IREF as supplied to the initialization routines G05YLF or G05YNF.

1: XMEAN(*idim*) – **double precision** array *Input*

On entry: specifies, for each dimension, the mean of the underlying Normal distribution.

Constraint: $|XMEAN(i)| \leq |-\log(X02AMF) - 10.0 \times STD(i)|$, for $i = 1, 2, \dots, idim$.

2: STD(*idim*) – **double precision** array *Input*

On entry: specifies, for each dimension, the standard deviation of the underlying Normal distribution.

Constraint: $STD(i) \geq 0$, for $i = 1, 2, \dots, idim$.

- 3: N – INTEGER *Input*
On entry: the number of quasi-random numbers required.
Constraint: $N \geq 0$.
- 4: QUASI(N,*idim*) – **double precision** array *Output*
On exit: contains the N quasi-random numbers of dimension *idim*.
- 5: IREF(*liref*) – INTEGER array *Communication Array*
On entry: contains information on the current state of the sequence.
On exit: contains updated information on the state of the sequence.
- 6: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, incorrect initialization has been detected.

IFAIL = 2

On entry, $N < 1$.

IFAIL = 3

On entry, at least one element of XMEAN is too large.

IFAIL = 4

There have been too many calls to the generator.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example calls G05YLF to initialize the generator and then G05YKF to produce a sequence of five four-dimensional quasi-random numbers variates.

9.1 Program Text

```

*      G05YKF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
      INTEGER          NOUT, IDIM, N
      PARAMETER        (NOUT=6,IDIM=4,N=5)
      INTEGER          LIREF
      PARAMETER        (LIREF=32*IDIM+7)
*      .. Local Scalars ..
      INTEGER          GENID, I, IFAIL, ISKIP, J
*      .. Local Arrays ..
      DOUBLE PRECISION QUASI(N,IDIM), STD(IDIM), XMEAN(IDIM)
      INTEGER          IREF(LIREF)
*      .. External Subroutines ..
      EXTERNAL         G05YKF, G05YLF
*      .. Intrinsic Functions ..
      INTRINSIC        DBLE
*      .. Executable Statements ..
      WRITE (NOUT,99999) 'G05YKF Example Program Results'
      IFAIL = 0
      DO 20 I = 1, IDIM
          XMEAN(I) = DBLE(I)
          STD(I) = 1.0D0
20  CONTINUE
*      Skip the first few variates in the sequence
      ISKIP = 1000
      IFAIL = 0
*      Initialize the Sobol generator
      GENID = 1
      CALL G05YLF(GENID,IDIM,IREF,LIREF,ISKIP,IFAIL)
*      Generate N values from the log-normal distribution
      CALL G05YKF(XMEAN,STD,N,QUASI,IREF,IFAIL)
      WRITE (NOUT,99998) ((QUASI(I,J),J=1,IDIM),I=1,N)
*
99999 FORMAT (1X,A,F20.4)
99998 FORMAT (1X,4F10.4)
      END

```

9.2 Program Data

None.

9.3 Program Results

```

G05YKF Example Program Results
  4.8648    9.4382    2.4979    21.5895
 17.7572    4.9813   41.8501   233.2386
  2.5195   20.5384   10.8353    45.3933
  1.8229    6.8823    6.9276    32.4808
  7.4938   49.7034   29.0198   127.4745

```
